



Progetto di Modellazione Concettuale per il Web Semantico

Survival International Ontology

Alessandro Contenti

A.A. 2021/2022

Table of Contents

1. Introduzione	1
2. Motivazioni	1
3. Requisiti	1
3.1. Finalità	1
3.2. Task	2
3.3. Utenti	2
4. Descrizione del dominio	3
4.1. Dominio	3
4.2. Riferimenti	3
5. Documentazione sul dominio	4
5.1. Allineamenti	8
6. LODE	9
7. Visualizzazione	9
7.1. Tassonomia delle classi	9
7.2. Template	10
7.3. Esempio	12
8. Queries SPARQL	13
8.1. Flow-chart	13
8.2. Mockups	14
8.3. Queries principali	17
8.4. Queries aggiuntive dell'applicazione client	23
9. Applicazione client	28

1. Introduzione

Il progetto svolto riguarda l'ambito della ONG "Survival International" e tutto il materiale prodotto è disponibile presso la [repository GitHub](#) del progetto stesso.

L'ontologia in formato Turtle è accesibile al seguente link
<https://acontenti.github.io/progetto-modsem/survival.ttl> (che è anche il suo IRI).

2. Motivazioni

Il dominio che ho deciso di modellare riguarda il contesto e le operazioni dell'associazione senza scopo di lucro "Survival International".

Survival International è un movimento internazionale per la salvaguardia dei diritti dei popoli indigeni. È stata fondata nel 1969 a Londra, a seguito della pubblicazione di un articolo del Sunday Times, che narrava il genocidio dei popoli amazzonici. La sua missione è aiutare i popoli indigeni e tribali a difendere le loro vite, le loro terre e i loro fondamentali diritti umani contro ogni forma di persecuzione, razzismo e genocidio.

A partire dalla sua fondazione, Survival ha portato avanti decine di campagne a favore dei popoli indigeni, ricevendo il sostegno di centinaia di migliaia di persone.

L'idea di modellare il dominio di Survival International mi è venuta leggendo casualmente un articolo riguardo questa associazione, in relazione alle loro campagne a supporto dei popoli incontattati, tema che mi incuriosisce molto.

3. Requisiti

3.1. Finalità

La finalità dello sviluppo di un'ontologia che descriva il contesto sia culturale che geografico in cui opera Survival, le operazioni che svolge e le criticità che deve affrontare è quello di creare una solida base di conoscenza che contenga tutte le informazioni utili per facilitare le operazioni interne dell'associazione e la comunicazione con il pubblico.

Nell'ontologia da me creata vi è solamente un sottoinsieme di tutte le

informazioni che potrebbero essere incluse in una base di conoscenza del genere.

Sia perché alcuni concetti potrebbero essere dettagliati con una granularità più fine.

Sia perché manca la modellazione di alcuni ambiti, come ad esempio la gestione dei contatti locali con i popoli indigeni e dei contatti governativi, la gestione dei fondi e molti altri aspetti su cui, da esterni, è impossibile ottenere informazioni.

3.2. Task

Alcuni task specifici dell'ontologia modellata sono:

- Gestire una lista aggiornata dei popoli indigeni che necessitano di aiuto da parte di Survival o che comunque sono oggetto d'interesse
- Gestire una lista aggiornata delle campagne in corso o passate e da chi sono gestite (per esempio se sono condivise con altre associazioni)
- Gestire una lista delle sedi locali di Survival e delle associazioni con cui collabora
- Cercare velocemente quali sono le minacce che mettono in pericolo uno specifico popolo indigeno e quali campagne sono in atto a loro supporto
- Localizzare facilmente i vari popoli per area geografica, visualizzando anche quali popoli condividono la stessa

3.3. Utenti

Gli utenti che potrebbero trarre giovamento dall'utilizzo di questa ontologia sono:

- **Personale di Survival:** per gestire e aggiornare la base di conoscenza e usarla per programmare e tenere sotto controllo le campagne e le operazioni da svolgere
- **Sostenitori di Survival:** per seguire le varie campagne in atto e la situazione dei vari popoli indigeni
- **Pubblico:** per scoprire le iniziative di Survival o informarsi riguardo i vari popoli indigeni

4. Descrizione del dominio

4.1. Dominio

Il dominio è composto da un insieme di concetti top-level, cui corrisponde una specifica classe nell'ontologia:

Tribù

Una tribù o un popolo indigeno oggetto dell'attenzione e delle campagne di Survival e/o di altre organizzazioni

Minaccia

Con minaccia si intende o un generico problema che affligge più popoli indigeni, accomunandoli come protagonisti di una campagna, oppure un problema che affligge una specifica tribù.

Campagna

Una campagna può essere diretta ad affrontare in generale una minaccia che riguarda molte comunità indigene, oppure si rivolge direttamente ad una specifica tribù, affrontandone le varie problematiche

Organizzazione

Un'organizzazione è un generico ente che collabora con Survival (come anche Survival stessa e le sue sedi locali). Un'organizzazione può essere legata a una campagna che conduce (o co-conduce).

Luogo

Un luogo indica in generale un'area geografica, più specificamente un continente, un paese o un habitat. Quest'ultimo rappresenta un'area in cui vivono una (o più) tribù e popoli indigeni, atta a circoscrivere più precisamente i territori in cui vivono queste popolazioni, che spesso si estendono oltre i confini di un singolo paese.

4.2. Riferimenti

La stragrande maggioranza delle informazioni riguardanti i temi chiave di questa ontologia, ovvero popoli indigeni, minacce, campagne di Survival, organizzazioni sono tratte dai siti web di Survival in inglese e italiano:

- Survival International: <https://www.survivalinternational.org>
- Survival International Italia: <https://www.survival.it>

Purtroppo le informazioni sono un po' frammentate e organizzate in modo poco sistematico. Probabilmente per fare più colpo sui lettori, piuttosto che per dare una documentazione precisa.

Alcune pagine non sembrano molto aggiornate, per cui è stato necessario integrare qualche volta con encyclopedie online come:

- Wikipedia: <https://en.wikipedia.org/> e <https://it.wikipedia.org/>
- Encyclopædia Treccani: <https://www.treccani.it/enciclopedia/>

Queste sono state utilizzate anche per trarre informazioni su concetti non specifici dell'ambito di Survival, come ad esempio continenti e paesi.

È stato inoltre consultato il sito dell'Articulation of Indigenous Peoples of Brazil (APIB): <https://apiboficial.org/>

5. Documentazione sul dominio

Come già riportato alla sezione precedente la maggioranza delle informazioni utilizzate per documentarsi sul dominio modellato deriva dai siti istituzionali di Survival International in italiano e inglese, che riportano circa le stesse informazioni nelle due lingue (il sito italiano non è sempre aggiornato come quello inglese). Quindi per semplicità mi riferisco ad entrambi indifferentemente come "il sito di Survival".

Il sito di Survival, appare poco strutturato e coerente: spesso alcune pagine evidentemente più vecchie hanno un'interfaccia diversa dalle altre.

Tuttavia presenta un minimo di suddivisione delle informazioni: è possibile infatti vedere una lista dei popoli indigeni seguiti da Survival, suddivisi per tre macroaeree: le Americhe, Africa e Asia ed Australasia.




Figure 1. Elenco di popoli indigeni

Inoltre è possibile vedere una lista (poco esaustiva) delle campagne in corso.




Figure 2. Piccola lista di alcune campagne di Survival

Aprendo ad esempio la pagina della campagna "Stop Brazil's Genocide" (inclusa nell'ontologia) ci si trova di fronte ad un lungo articolo, molto interessante, ma totalmente privo di informazioni strutturate: senza leggerlo per intero è impossibile capire al volo quali popoli siano interessati da questa campagna, chi vi collabori, se sia ancora in corso o terminata.

Leggendo bene fino a circa metà pagina, si nota che è una campagna in cui Survival collabora con l'APIB, l'*Associazione dei Popoli Indigeni del Brasile* (anch'essa inclusa nell'ontologia).



The screenshot shows the homepage of Survival International. At the top, there's a black navigation bar with the word "Survival" and a white feather logo on the left. To the right are links for "What we do", "About us", "Tribes & campaigns", "News", "Shop", and language options: Deutsch, Español, Français, Italiano, Nederlands, Português, Hindi, Odia, and a search icon. Below the navigation is a teal button labeled "Get involved" and a red button labeled "Donate now". The main headline reads "Across Brazil, indigenous peoples are fighting back" in bold black text. Below the headline is a large photograph of a protest. In the photo, several indigenous men are visible, some shirtless and painted with traditional markings, holding wooden staffs. A child is seen in the foreground. In the background, there are flags, including the Brazilian flag, and other protesters. The photo is credited to "© Sarah Shenker/Survival International". Below the photo, there are three paragraphs of text from the APIB. The first paragraph quotes APIB as saying: "We have the right to exist. We won't retreat. We'll denounce this government around the world." The second paragraph states: "Brazil's indigenous peoples, their supporters and allies, and Survival International are fighting back as the President plots to steal their land, destroy their lives and annihilate them as peoples." The third paragraph concludes: "The biggest ever international protest for indigenous rights took place in Bolsonaro's first month in office, and raising our voices together has already make a difference."

Across Brazil, indigenous peoples are fighting back

© Sarah Shenker/Survival International

APIB, the Association of Indigenous Peoples of Brazil, said: "We have the right to exist. We won't retreat. We'll denounce this government around the world."

Brazil's indigenous peoples, their supporters and allies, and Survival International are fighting back as the President plots to steal their land, destroy their lives and annihilate them as peoples.

The biggest ever international protest for indigenous rights took place in Bolsonaro's first month in office, and raising our voices together has already make a difference.

Figure 3. Pagina della campagna "Stop Brazil's Genocide"

Anche il sito web dell'associazione APIB sopracitata è stato utilizzato per documentarsi, ma è ancora meno strutturato e caotico (e presenta anche qualche bug).

Le informazioni sono sparse su varie pagine poco chiaramente accessibili e talvolta disponibili solo in portoghese.



Figure 4. Sito web dell'APIB

Data la scarsità di informazioni pratiche, in favore di "propaganda" e richieste di donazioni o firme per la causa (fatto comprensibilissimo, ma in questo caso spiacente), è stato talvolta necessario documentarsi su Wikipedia.

Yanomamö

Da Wikipedia, l'encyclopédie libre.

Questa voce o sezione sull'argomento antropologia è priva o carente di note e riferimenti bibliografici puntuali.

Sebbene vi siano una **bibliografia** e/o dei **collegamenti esterni**, manca la contestualizzazione delle fonti con **note a piè di pagina** o altri riferimenti precisi che indichino puntualmente la provenienza delle informazioni. Puoi **migliorare questa voce citando le fonti** più precisamente.

Gli yanomamö sono un **gruppo etnico** che abita per lo più la zona di foresta compresa tra i bacini dei fiumi **Orinoco** e **Rio delle Amazzoni**, sicché la frontiera tra il **Venezuela** e il **Brasile** attraversa il loro territorio tradizionale. Si stima che abbiano una **popolazione** di circa 27.000-32.000 individui^[1]. Denominazioni alternative sono Yanomame, Yanomami, Guica, Guaharibo, Shamatari, Cobari Kobali, Cobariwa.

L'invasione delle loro terre su grande scala ebbe inizio intorno al 1970 con la costruzione di strade e proseguì, intensificandosi alla fine degli anni ottanta con l'inizio della corsa all'oro in Amazzonia. Prima dell'arrivo dei **garimpeiros** (cercatori d'oro) vi erano stati contatti solamente da parte di **missionari** cattolici e protestanti, e sporadicamente con cacciatori o raccoglitori di gomma. L'incontro con i garimpeiros fu particolarmente rilevante perché diede luogo a violenze da parte di questi ultimi, che ebbero anche rilevanza internazionale in materia di violazione dei diritti umani (l'eccidio di una ventina di yanomami, soprattutto bambini e donne nel 1993^[2]).

Indice [nascondi]

- 1 Stile di vita
- 2 Educazione
- 3 Insediamenti
- 4 Lingua
- 5 Note
- 6 Bibliografia
- 7 Voci correlate
- 8 Altri progetti
- 9 Collegamenti esterni

Yanomamö

Luogo d'origine Venezuela, Brasile

Popolazione 27.000 - 32.000 individui

Lingua yanomamö

Religione animismo

Figure 5. Pagina Wikipedia del popolo Yanomamö, interessato dalla campagna precedente

In conclusione, date le note negative espresse sul sito di Survival e altre associazioni analoghe, l'idea di fondo di questa ontologia e dell'applicazione client derivante è quella di permettere agli utenti di avere informazioni chiare

e facilmente accessibili, il più possibili ricche di dettagli, piuttosto che di appelli.

Sarebbe molto interessante poter unire i due aspetti in un possibile "nuovo" sito web di Survival che presenti tutti i contenuti già visibili, ma ben collegati semanticamente e con allegate schede che illustrino in breve le informazioni chiave.

Così da non trascurare l'informatività del sito, pur mantenendo la sua funzione giustamente "propagandistica".

5.1. Allineamenti

L'ontologia sviluppata è stata allineata sia con [Schema.org](#) che con [Wikidata](#).

Con *Schema.org* sono state allineate solo alcune classi e proprietà che rappresentano concetti o relazioni di alto livello, poiché non vi è molta sovrapposizione tra le due ontologie.

Classi:

- `survival:Location` → `schema:Place`
- `survival:Continent` → `schema:Continent`
- `survival:Country` → `schema:Country`
- `survival:Organization` → `schema:Organization`

Proprietà:

- `survival:locatedIn` → `schema:location`
- `survival:branchOf` → `schema:parentOrganization`
- `survival:hasBranch` → `schema:subOrganization`

Con *Wikidata*, grazie alla sua grande varietà, invece sono state allineate non solo molte classi e proprietà, ma anche moltissimi individui.

Dato il grande numero non è possibile riportare tutti gli allineamenti, per cui saranno riportati solo alcuni esempi significativi.

Classi:

- `survival:Tribe` → `wd:Q103817 (indigenous people)`
- `survival:Campaign` → `wd:Q18812548 (campaign)`
- `survival:Organization` → `wd:Q43229 (organization)`

Proprietà:

- `survival:locatedIn` → `wdt:P706` (*located in/on physical feature*)
- `survival:branchOf` → `wdt:P749` (*parent organization*)

Individui:

- `survival:SurvivalInternational` → `wd:Q1887464` (*Survival International*)
- `survival:APIB` → `wd:Q16494254` (*Brazil's Indigenous People Articulation*)
- `survival:NorthSentinel` → `wd:Q1260331` (*North Sentinel Island*)
- `survival:Sentinelese` → `wd:Q917695` (*Sentinelese people*)

Tutti gli allineamenti delle classi sono stati effettuati con la proprietà `owl:equivalentClass` perché sono state scelte classi che rappresentassero esattamente lo stesso concetto nelle varie ontologie.

Per quanto riguarda le proprietà si è utilizzato `owl:equivalentProperty` per tutte le proprietà strettamente equivalenti, tranne in qualche caso in cui si è utilizzato `rdfs:subPropertyOf` per proprietà che sussumessero la proprietà importata.


Per gli tutti individui si è naturalmente utilizzato `owl:sameAs`.

6. LODE

La documentazione generata con LODE (per la precisione con il tool WIDOCO che utilizza LODE) è disponibile al seguente indirizzo web:
<https://acontegni.github.io/progetto-modsem/index-en.html>

7. Visualizzazione

7.1. Tassonomia delle classi



7.2. Template

Durante lo sviluppo dell'ontologia si è deciso di utilizzare gli Ontology Design Patterns ["Participation"](#) e ["Co-participation"](#).




Figure 6. Diagramma dell'ODP "Participation"

L'ODP Participation è stato utilizzato per differenziare gli oggetti dagli eventi e stabilire la partecipazione degli oggetti agli eventi.

Nello specifico la classe **survival:Campain** è stata resa sottoclasse di **Event** mentre tutte le altre classi sono state resse sottoclassi di **Object**.

Inoltre, la proprietà **survival:leads**, che mette in relazione **survival:Organization** con **survival:Campain**, è stata resa sottoproprietà di **isParticipantIn**; lo stesso per la proprietà inversa **survival:ledBy** con **hasParticipant**.




Figure 7. Diagramma dell'ODP "Co-participation"

Quindi, grazie all'ODP Co-participation (che estende l'ODP Participation), si è modellata la proprietà simmetrica **survival:collaboratesWith** tra due **survival:Organization**, come sottoproprietà di **coparticipatesWith**. Ciò indica che due oggetti, in questo caso due organizzazioni, collaborano in un evento, in questo caso una campagna, se entrambe partecipano ad uno stesso evento.

7.3. Esempio




Figure 8. Grafico di esempio di utilizzo della proprietà `survival:collaboratesWith`

	subject	predicate	object	context
1	survival:collaboratesWith	rdf:type	owl:ObjectProperty	http://www.ontotext.com/explicit
2	survival:collaboratesWith	rdf:type	owl:SymmetricProperty	http://www.ontotext.com/explicit
3	survival:collaboratesWith	rdfs:comment	"Links an organization with another organization it collaborates with in one or more campaigns"@en	http://www.ontotext.com/explicit
4	survival:collaboratesWith	rdfs:domain	survival:Organization	http://www.ontotext.com/explicit
5	survival:collaboratesWith	rdfs:label	"Collaborates with"@en	http://www.ontotext.com/explicit
6	survival:collaboratesWith	rdfs:label	"Collabora con"@it	http://www.ontotext.com/explicit
7	survival:collaboratesWith	rdfs:range	survival:Organization	http://www.ontotext.com/explicit
8	survival:collaboratesWith	rdfs:subPropertyOf	coparticipation:coparticipatesWith	http://www.ontotext.com/explicit
9	survival:collaboratesWith	owl:propertyChainAxiom	_:node242	http://www.ontotext.com/explicit

Figure 9. Triple che descrivono la proprietà `survival:collaboratesWith`

	subject	predicate	object	context
1	survival:APIB	survival:collaboratesWith	survival:APIB	http://www.ontotext.com/implicit
2	survival:APIB	survival:collaboratesWith	survival:SurvivalInternational	http://www.ontotext.com/implicit
3	survival:SurvivalInternational	survival:collaboratesWith	survival:APIB	http://www.ontotext.com/implicit
4	survival:SurvivalInternational	survival:collaboratesWith	survival:SurvivalInternational	http://www.ontotext.com/implicit

Figure 10. Triple che descrivono un esempio di utilizzo della proprietà `survival:collaboratesWith`

8. Queries SPARQL

Qui di seguito è presente il materiale riguardante la sezione delle queries SPARQL.

Sono state create 5 query principali e 5 query accessorie.

8.1. Flow-chart

Il flow-chart seguente descrive lo schema di interazione dell'applicazione nel complesso.

Sono rappresentate tutte le pagine e come nodi del grafo, mentre le frecce indicano la possibilità di navigare da una pagina all'altra.

Per evitare di appesantire eccessivamente il grafico, si sono evitate le frecce che collegano tra loro le pagine "top-level", cioè quelle accessibili dalla barra di navigazione, proprio perché sono sempre accessibili da qualsiasi pagina.




Figure 11. Flow-chart dell'interazione

8.2. Mockups

I seguenti mockup illustrano alcuni esempi di pagine del progetto dell'applicazione client.

Oltre alla pagina iniziale, vi sono fondamentalmente due tipologie di pagine: quelle che presentano una lista di elementi, come la seconda e la terza, e quelle che rappresentano il dettaglio di un singolo elemento, come le ultime due.





Figure 12. Pagina iniziale

The screenshot shows the 'Tribes' page of the Survival International Ontology Explorer. The vertical sidebar on the left has 'Tribes' selected and highlighted in blue. The main content area displays a table with six rows, each representing a tribe, its habitat, and whether it has been contacted or not.

Tribe	Habitat	Contacted
Mashco-Piro	The Amazon Uncontacted Frontier	no
Matsés	The Amazon Uncontacted Frontier	no
San	Central Kalahari Game Reserve	yes
Sentinelese	North Sentinel Island	no
Yanomamö	Yanomami Park	no


Figure 13. Pagina "Tribes"



The screenshot shows a web browser window with the URL <https://www.sioex.org/>. The sidebar on the left has a blue highlight over the 'Countries' link. The main content area displays a table with data about tribes across different countries.

Country	Continent	Tribes
Brazil	South America	3
Bolivia	South America	2
Botswana	Africa	1
India	Asia	1
Italy	Europe	0
Peru	South America	2
United Kingdom	Europe	0

Figure 14. Pagina "Countries"



The screenshot shows a web browser window with the URL <https://www.sioex.org/>. The sidebar on the left has a blue highlight over the 'Countries' link. The main content area displays a table with details about the Sentinelese tribe.

Tribe:	Sentinelese
Description:	The Sentinelese are an uncontacted tribe living on North Sentinel Island, one of the Andaman Islands in the Indian Ocean. They vigorously reject all contact with outsiders.
Contacted:	no
Habitats:	North Sentinel Island
Countries:	India
Threats:	Forced contact, Illegal fishing
Campaigns:	Uncontacted Tribes

Figure 15. Pagina "Tribe" per l'individuo "survival:Sentinelese"

Organization:	Survival International
Description:	Survival International is a human rights organisation formed in 1969 that campaigns for the rights of indigenous and/or tribal peoples and uncontacted peoples.
Website:	https://www.survivalinternational.org
E-mail:	info@survivalinternational.org
Parent organization:	-
Branches:	Survival International Brazil Survival International Italy
Countries:	United Kingdom
Campaigns:	Stop Brazil's genocide Decolonize Conservation Uncontacted Tribes

Figure 16. Pagina "Organization" per l'individuo "survival:SurvivalInternational"

8.3. Queries principali

Queste sono le query principali dell'applicazione client e sono accessibili tramite il menù di navigazione

8.3.1. Tribes

Questa query restituisce la lista dei popoli indigeni presenti nell'ontologia e, per ogni popolo, restituisce anche il nome, se è contattato o meno e l'habitat in cui vive con il relativo nome.

```

PREFIX survival: <https://acontenti.github.io/progetto-modsem/survival.ttl#>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX onto: <http://www.ontotext.com/>

SELECT ?tribe ?name ?contacted ?habitat (MAX(?hLabel) as ?habitatLabel)
FROM onto:disable-sameAs
WHERE {
    ?tribe rdf:type survival:Tribe;
            survival:name ?name;
            survival:contacted ?contacted;
            survival:livesIn ?habitat.
    ?habitat rdfs:label ?hLabel.
    FILTER(LANG(?hLabel) = "" || LANGMATCHES(LANG(?hLabel), "en"))
}
GROUP BY ?tribe ?name ?contacted ?habitat
ORDER BY ?name

```

	tribe	name	contacted	habitat	habitatLabel
1	survival:MashcoPiro	"Mashco-Piro"	"false"^^xsd:boolean	survival:UncontactedFrontier	"The Amazon Uncontacted Frontier"@en
2	survival:Matses	"Matsés"	"false"^^xsd:boolean	survival:UncontactedFrontier	"The Amazon Uncontacted Frontier"@en
3	survival:Bushmen	"San"	"true"^^xsd:boolean	survival:CentralKalahariGame	"Central Kalahari Game Reserve"@en
4	survival:Sentinelese	"Sentinelese"	"false"^^xsd:boolean	survival:NorthSentinel	"North Sentinel Island"@en
5	survival:Yanomami	"Yanomamō"	"false"^^xsd:boolean	survival:YanomamiPark	"Yanomami Park"@en

Figure 17. Esempio di risultato della query

8.3.2. Countries

Questa query restituisce la lista dei paesi presenti nell'ontologia e per ciascuno restituisce anche il relativo nome, il nome del continente in cui si trova e il numero di popoli indigeni che vi risiedono.

```

PREFIX survival: <https://acontenti.github.io/progetto-modsem/survival.ttl#>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX onto: <http://www.ontotext.com/>

SELECT ?country ?countryLabel ?continentLabel ?tribesCount
FROM onto:disable-sameAs
WHERE {
    ?country rdf:type survival:Country;
              rdfs:label ?countryLabel;
              survival:locatedIn ?continent.
    ?continent rdfs:label ?continentLabel.
    OPTIONAL {
        SELECT ?country (COUNT(?tribe) as ?count)
        WHERE {
            ?tribe rdf:type survival:Tribe;
                    survival:livesIn ?habitat.
            ?habitat rdf:type survival:Habitat;
                    survival:locatedIn ?country.
        } GROUP BY ?country
    }
    FILTER(LANG(?countryLabel) = "" || LANGMATCHES(LANG(?countryLabel), "en"))
    FILTER(LANG(?continentLabel) = "" || LANGMATCHES(LANG(?continentLabel),
    "en"))
    BIND(COALESCE(?count, 0) AS ?tribesCount)
}

```

	country	countryLabel	continentLabel	tribesCount
1	survival:Brazil	"Brazil"@en	"South America"@en	"3"^^xsd:integer
2	survival:Bolivia	"Bolivia"@en	"South America"@en	"2"^^xsd:integer
3	survival:Botswana	"Botswana"@en	"Africa"@en	"1"^^xsd:integer
4	survival:India	"India"@en	"Asia"@en	"1"^^xsd:integer
5	survival:Italy	"Italy"@en	"Europe"@en	"0"^^xsd:integer
6	survival:Peru	"Peru"@en	"South America"@en	"2"^^xsd:integer
7	survival:UK	"United Kingdom"@en	"Europe"@en	"0"^^xsd:integer

Figure 18. Esempio di risultato della query

8.3.3. Campaigns

Questa query restituisce la lista delle campagne presenti nell'ontologia e, per ogni campagna, restituisce anche il relativo nome, l'URL della pagina web della campagna e il numero di popoli indigeni aiutati da essa.

```

PREFIX survival: <https://acontenti.github.io/progetto-modsem/survival.ttl#>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX onto: <http://www.ontotext.com/>

SELECT ?campaign ?campaignName ?url (COUNT(?tribe) AS ?tribesCount)
FROM onto:disable-sameAs
WHERE {
    ?campaign rdf:type survival:Campaign;
        rdfs:label ?campaignName;
        survival:campaignPage ?url;
        survival:helps ?tribe.
    FILTER(LANG(?campaignName) = "" || LANGMATCHES(LANG(?campaignName), "en"))
}
GROUP BY ?campaign ?campaignName ?url
ORDER BY ?campaignName

```

	campaign	campaignName	url	tribesCount
1	survival:DecolonizeConservation	"Decolonize Conservation"@en	"https://www.survivalinternational.org/conservation"^^xsd:anyURI	"1"^^xsd:integer
2	survival:StopBrazilGenocide	"Stop Brazil's genocide"@en	"https://www.survivalinternational.org/brazilgenocide"^^xsd:anyURI	"1"^^xsd:integer
3	survival:UncontactedTribes	"Uncontacted Tribes"@en	"https://www.survivalinternational.org/uncontactedtribes"^^xsd:anyURI	"3"^^xsd:integer

Figure 19. Esempio di risultato della query

8.3.4. Organizations

Questa query restituisce la lista delle organizzazioni presenti nell'ontologia e per ciascuna restituisce anche il nome e l'organizzazione cui fa capo (se ne ha una) con il relativo nome.

```

PREFIX survival: <https://acontenti.github.io/progetto-modsem/survival.ttl#>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX owl: <http://www.w3.org/2002/07/owl#>
PREFIX onto: <http://www.ontotext.com/>

SELECT ?org (MAX(?label) as ?orgLabel) ?head (MAX(?hLabel) as ?headLabel)
FROM onto:disable-sameAs
WHERE {
    ?org rdf:type survival:Organization;
          rdfs:label ?label.
    OPTIONAL {
        ?org survival:branchOf ?head.
        ?head rdfs:label ?hLabel.
        FILTER(LANG(?hLabel) = "" || LANGMATCHES(LANG(?hLabel), "en"))
    }
    FILTER(LANG(?label) = "" || LANGMATCHES(LANG(?label), "en"))
}
GROUP BY ?org ?head
ORDER BY ?orgLabel

```

	org	orgLabel	head	headLabel
1	survival:APIB	"Brazil's Indigenous People Articulation"@en		
2	survival:SurvivalInternational	"Survival International"@en		
3	survival:SurvivalBrasil	"Survival International Brazil"@en	survival:SurvivalInternational	"Survival International"@en
4	survival:SurvivalItalia	"Survival International Italy"@en	survival:SurvivalInternational	"Survival International"@en

Figure 20. Esempio di risultato della query

8.3.5. Tribes threats

Questa query restituisce la lista dei popoli indigeni presenti nell'ontologia e, per ogni popolo, restituisce anche il relativo nome, le minacce cui è sottoposto (sotto forma di concatenazione di stringhe).

Inoltre per ogni popolo restituisce un "threat score", ovvero un punteggio da 0 a 1 che esprime quanto il popolo in questione sia in pericolo, calcolato sul numero di minacce subite.

```

PREFIX survival: <https://acontenti.github.io/progetto-modsem/survival.ttl#>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX onto: <http://www.ontotext.com/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

SELECT ?tribe ?name ?threats (?count/?max AS ?threatScore)
FROM onto:disable-sameAs
WHERE {
  {
    SELECT (MAX(?count) AS ?max)
    WHERE {
      SELECT ?tribe (COUNT(?threat) as ?count)
      WHERE {
        ?tribe survival:threatenedBy ?threat.
      }
      GROUP BY ?tribe
    }
  }
  {
    SELECT ?tribe ?name (COUNT(?threat) as ?count)
    (GROUP_CONCAT(?threatLabel;separator=', ') AS ?threats)
    WHERE {
      ?tribe rdf:type survival:Tribe;
              survival:name ?name;
              survival:threatenedBy ?threat.
      ?threat rdfs:label ?threatLabel.
      FILTER(LANG(?threatLabel) = "" || LANGMATCHES(LANG(?threatLabel),
      "en"))
    }
    GROUP BY ?tribe ?name
  }
}

```

	tribe	name	threats	threatScore
1	survival:Bushmen	"San"	"Colonial conservation"	"0.5"^^xsd:decimal
2	survival:MashcoPiro	"Mashco-Piro"	"Deforestation, Forced contact"	"1"^^xsd:decimal
3	survival:Yanomami	"Yanomamô"	"Deforestation, Illegal mining"	"1"^^xsd:decimal
4	survival:Matses	"Matsés"	"Forced contact"	"0.5"^^xsd:decimal
5	survival:Sentinelese	"Sentinelese"	"Forced contact, Illegal fishing"	"1"^^xsd:decimal

Figure 21. Esempio di risultato della query

8.4. Queries aggiuntive dell'applicazione client

Le seguenti query sono utilizzate dall'applicazione client per estrarre dettagli e informazioni utili per un singolo individuo dell'ontologia.

8.4.1. Tribe

Questa query, dato l'IRI di un popolo indigeno (`survival:Tribe`), ne restituisce il nome, la descrizione, se è contattato, le minacce cui è sottoposto, gli habitat in cui vive e i paesi in cui vive (inferito dagli habitat)

```

PREFIX survival: <https://acontenti.github.io/progetto-modsem/survival.ttl#>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX onto: <http://www.ontotext.com/>

PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
SELECT ?name ?comment ?contacted ?threats ?habitat ?habitatLabel ?country
?countryLabel ?campaign ?campaignLabel
FROM onto:disable-sameAs
WHERE {
{
    SELECT ?tribe ?name ?comment ?contacted (GROUP_CONCAT(DISTINCT
?threatLabel;separator=', ') AS ?threats) ?habitat (MAX(?hLabel) AS
?habitatLabel) ?country (MAX(?cLabel) AS ?countryLabel) WHERE {
        BIND(<${id}> AS ?tribe)
        ?tribe survival:name ?name;
            rdfs:comment ?comment;
            survival:contacted ?contacted;
            survival:livesIn ?habitat.
        FILTER(LANG(?comment) = "" || LANGMATCHES(LANG(?comment), "en"))
        OPTIONAL {
            ?tribe survival:threatenedBy ?threat.
            ?threat rdfs:label ?threatLabel.
            FILTER(LANG(?threatLabel) = "" ||
LANGMATCHES(LANG(?threatLabel), "en"))
        }
        OPTIONAL {
            ?tribe survival:livesIn ?habitat.
            ?habitat rdfs:label ?hLabel.
            FILTER(LANG(?hLabel) = "" || LANGMATCHES(LANG(?hLabel), "en"))
        }
        OPTIONAL {
            ?habitat survival:locatedIn ?country.
            ?country rdf:type survival:Country;
                rdfs:label ?cLabel.
            FILTER(LANG(?cLabel) = "" || LANGMATCHES(LANG(?cLabel), "en"))
        }
    } GROUP BY ?tribe ?name ?comment ?contacted ?habitat ?country
}
OPTIONAL {
    ?tribe survival:helpedBy ?campaign.
    ?campaign rdfs:label ?campaignLabel.
    FILTER(LANG(?campaignLabel) = "" || LANGMATCHES(LANG(?campaignLabel),
"en"))
}
}
}

```

8.4.2. Country

Questa query, dato l'IRI di un paese (**survival:Country**), ne restituisce il nome, il continente in cui è situato, gli habitat che contiene e le organizzazioni che vi hanno sede

```
PREFIX survival: <https://acontenti.github.io/progetto-modsem/survival.ttl#>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX onto: <http://www.ontotext.com/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

SELECT (MAX(?label) AS ?name) (MAX(?cLabel) AS ?continentLabel) ?habitat
(MAX(?hLabel) AS ?habitatLabel) ?tribe ?tribeLabel ?org (MAX(?oLabel) AS
?orgLabel)
FROM onto:disable-sameAs
WHERE {
    BIND(<${id}> AS ?country)
    ?country rdfs:label ?label.
    FILTER(LANG(?label) = "" || LANGMATCHES(LANG(?label), "en"))
    ?country survival:locatedIn ?continent.
    ?continent rdfs:label ?cLabel.
    FILTER(LANG(?cLabel) = "" || LANGMATCHES(LANG(?cLabel), "en"))
    OPTIONAL {
        ?habitat rdf:type survival:Habitat;
                  survival:locatedIn ?country;
                  rdfs:label ?hLabel.
        FILTER(LANG(?hLabel) = "" || LANGMATCHES(LANG(?hLabel), "en"))
        OPTIONAL {
            ?tribe rdf:type survival:Tribe;
                    survival:livesIn ?habitat;
                    survival:name ?tribeLabel.
        }
    }
    OPTIONAL {
        ?org rdf:type survival:Organization;
              survival:locatedIn ?country;
              rdfs:label ?oLabel.
        FILTER(LANG(?oLabel) = "" || LANGMATCHES(LANG(?oLabel), "en"))
    }
} GROUP BY ?habitat ?tribe ?tribeLabel ?org
```

8.4.3. Habitat

Questa query, dato l'IRI di un habitat (**survival:Habitat**), ne restituisce il nome,

la descrizione, la tipologia, i paesi in cui si trova e i popoli indigeni che vi risiedono

```
PREFIX survival: <https://acontenti.github.io/progetto-modsem/survival.ttl#>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX onto: <http://www.ontotext.com/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

SELECT (MAX(?label) AS ?name) ?comment ?type ?country (MAX(?cLabel) AS
?countryLabel) ?tribe ?tribeLabel
FROM onto:disable-sameAs
WHERE {
    BIND(<${id}> AS ?habitat)
    ?habitat rdfs:label ?label;
        rdfs:comment ?comment;
        survival:habitatType ?type.
    FILTER(LANG(?label) = "" || LANGMATCHES(LANG(?label), "en"))
    FILTER(LANG(?comment) = "" || LANGMATCHES(LANG(?comment), "en"))
    OPTIONAL {
        ?habitat survival:locatedIn ?country.
        ?country rdf:type survival:Country;
            rdfs:label ?cLabel.
        FILTER(LANG(?cLabel) = "" || LANGMATCHES(LANG(?cLabel), "en"))
    }
    OPTIONAL {
        ?tribe rdf:type survival:Tribe;
            survival:livesIn ?habitat;
            survival:name ?tribeLabel.
    }
} GROUP BY ?comment ?type ?country ?tribe ?tribeLabel
```

8.4.4. Organization

Questa query, dato l'IRI di un'organizzazione (`survival:Organization`), ne restituisce il nome, la descrizione, il sito web, l'e-mail, il paese in cui ha sede, le campagne che conduce, le sue eventuali sedi ed eventualmente l'organizzazione a cui fa capo

```

PREFIX survival: <https://acontenti.github.io/progetto-modsem/survival.ttl#>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX onto: <http://www.ontotext.com/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

SELECT (MAX(?label) AS ?name) ?comment ?website ?email ?country (MAX(?cLabel)
AS ?countryLabel) ?campaign ?campaignLabel ?branch (MAX(?bLabel) AS
?branchLabel) ?head (MAX(?hLabel) AS ?headLabel)
FROM onto:disable-sameAs
WHERE {
    BIND(<${id}> AS ?org)
    ?org rdfs:label ?label;
        rdfs:comment ?comment;
        survival:website ?website;
        survival:email ?email.
    FILTER(LANG(?label) = "" || LANGMATCHES(LANG(?label), "en"))
    FILTER(LANG(?comment) = "" || LANGMATCHES(LANG(?comment), "en"))
    OPTIONAL {
        ?org survival:locatedIn ?country.
        ?country rdf:type survival:Country;
            rdfs:label ?cLabel.
        FILTER(LANG(?cLabel) = "" || LANGMATCHES(LANG(?cLabel), "en"))
    }
    OPTIONAL {
        ?campaign rdf:type survival:Campaign;
            survival:ledBy ?org;
            rdfs:label ?campaignLabel.
        FILTER(LANG(?campaignLabel) = "" || LANGMATCHES(LANG(?campaignLabel), "en"))
    }
    OPTIONAL {
        ?branch survival:branchOf ?org;
            rdfs:label ?bLabel.
        FILTER(LANG(?bLabel) = "" || LANGMATCHES(LANG(?bLabel), "en"))
    }
    OPTIONAL {
        ?head survival:hasBranch ?org;
            rdfs:label ?hLabel.
        FILTER(LANG(?hLabel) = "" || LANGMATCHES(LANG(?hLabel), "en"))
    }
} GROUP BY ?comment ?website ?email ?country ?campaign ?campaignLabel ?branch
?head

```

8.4.5. Campaign

Questa query, dato l'IRI di una campagna (`survival:Campagne`), ne restituisce il nome, la descrizione, il sito web, i popoli che aiuta e le organizzazioni che la conducono

```
PREFIX survival: <https://acontenti.github.io/progetto-modsem/survival.ttl#>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX onto: <http://www.ontotext.com/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

SELECT (MAX(?label) AS ?name) ?comment ?website ?tribe (MAX(?tLabel) AS
?tribeLabel) ?org (MAX(?oLabel) AS ?orgLabel)
FROM onto:disable-sameAs
WHERE {
    BIND(<${id}> AS ?campaign)
    ?campaign rdfs:label ?label;
        rdfs:comment ?comment;
        survival:campaignPage ?website.
    FILTER(LANG(?label) = "" || LANGMATCHES(LANG(?label), "en"))
    FILTER(LANG(?comment) = "" || LANGMATCHES(LANG(?comment), "en"))
    OPTIONAL {
        ?tribe rdf:type survival:Tribe;
            survival:helpedBy ?campaign;
            rdfs:label ?tLabel.
        FILTER(LANG(?tLabel) = "" || LANGMATCHES(LANG(?tLabel), "en"))
    }
    OPTIONAL {
        ?org rdf:type survival:Organization;
            survival:leads ?campaign;
            rdfs:label ?oLabel.
        FILTER(LANG(?oLabel) = "" || LANGMATCHES(LANG(?oLabel), "en"))
    }
} GROUP BY ?website ?comment ?tribe ?org
```

9. Applicazione client

L'applicazione client, intitolata *Survival International Ontology Explorer*, è stata realizzata con il framework JavaScript *Quasar*, costruito a sua volta sul framework *Vue.js*.

Ho scelto questo framework perché permette di costruire velocemente una single page application (SPA) ben strutturata, senza dover creare tutta l'architettura e la UI da zero, dando quindi allo sviluppatore la possibilità di

concentrarsi sul contenuto invece che sulla forma.

L'applicazione rispecchia a grandi linee quanto illustrato nei mockup presenti nella sezione precedente e permette di visualizzare i dati delle queries viste, in maniera totalmente trasparente per l'utente.

Le sezioni dell'applicazione, che rappresentano le varie queries, sono accessibili dalla pagina iniziale e dal menù di navigazione (a sinistra).

Inoltre sono presenti le pagine che permettono di visualizzare (grazie alle query accessorie alla sezione precedente) i dettagli dei singoli individui appartenenti alle seguenti classi:

- **survival:Campaign** → Campagna
- **survival:Country** → Paese
- **survival:Habitat** → Habitat
- **survival:Organization** → Organizzazione
- **survival:Tribe** → Popolo indigeno

Queste ultime non sono accessibili direttamente tramite la barra di navigazione, bensì dalle varie sezioni dell'applicazione.

Infatti in ogni sezione alcuni elementi dei risultati, a seconda della specifica sezione, sono dotati di link che indirizzano alla pagina di dettaglio dello specifico individuo.




Figure 22. Schermata iniziale dell'applicazione client